Subject: Gambia DHS 2013
Posted by Ceesay1 on Tue, 08 Nov 2016 09:43:11 GMT
View Forum Message <> Reply to Message

Hello Everyone,
I have downloaded the Gambia's DHS2013 and currently trying to analysis it. I am interested in
the predictors of HIV prevalence in the Gambia.
To do that, I first merge the GMIR with the GMAR datasets to get the HIV testing results for
women and I similarly merged GMMR and the the GMAR datasets. The two newly created
datasets were appended to hive the hivtests datasets for men and women. For example:
1.In the men dataset, I did:
keep mv001 mv002 mv003 mv012
rename mv001 v001
rename mv002 v002
rename mv003 v003
rename mv012 v012
gen long id=((1000+v001)*10000)+(v002*100)+v003
gen gender="men"
2.Women dataset:
keep v001 v002 v003 v012
gen long id=((1000+v001)*10000)+(v002*100)+v003
gen gender="women"
I appended the above two which give rise to a dataset( Gender included). For the HIV test
dataset, I kept id hiv03 hiv05.
Finally merged the the new dataset which has gender included with HIV dataset:
use "C:\Users\IHP0015\Desktop\Gambia_Stata\menwomen.DTA", clear
merge 1:1 id using "C:\Users\IHP0015\Desktop\Gambia_Stata\hiv.DTA"

The next attempt is to attempt to replicate some of the tables in the Gambia DHS report but to my
dismay, all of the figures I produced were close but different.
Does anyone have an idea for the way forward?
Thanks

Posted by Liz-DHS on Fri, 11 Nov 2016 14:19:05 GMT
View Forum Message <> Reply to Message

A response from Dr. Tom Pullum,
Quote:
Rather than comment on your steps I will just give you the lines that will accomplish what you
want to do. You will need to modify the paths. I am using the old version of the merge command,
which for me is less error prone than the new version. Note that in the merged file you should use
hiv05 as the weight.
set more off
use e:\DHS\DHS_data\AR_files\GMAR60FL.dta, clear
rename hivclust v001
rename hivnumb v002
rename hivline v003
drop hiv01 hiv02
sort v001 v002 v003
save e:\DHS\DHS_data\scratch\GMARtemp.dta, replace
use e:\DHS\DHS_data\IR_files\GMIR60FL.dta, clear
keep v*
gen sex=2
save e:\DHS\DHS_data\scratch\GMIRtemp.dta, replace
use e:\DHS\DHS_data\MR_files\GMMR60FL.dta, clear
keep mv*
rename mv* v*
gen sex=1
append using e:\DHS\DHS_data\scratch\GMIRtemp.dta
sort v001 v002 v003
merge v001 v002 v003 using e:\DHS\DHS_data\scratch\GMARtemp.dta
tab _merge
keep if _merge==3
tab sex hiv03 [iweight=hiv05/1000000], row

Subject: Re: Gambia DHS 2013
Posted by Ceesay1 on Wed, 23 Nov 2016 13:18:31 GMT
View Forum Message <> Reply to Message

Thank you for the swift response. It was a useful and helpful suggestion.
In an attempt to replicate the tables, I was faced with a some unsolved issues. After invoking the
svyset command, I perfectly got all the results on table 1. A good replication.

But, I came across a number of problems with table 2.For instance the footnote of table 2
indicated that 2 cases had no information on religion and 4 cases of missing information on
religion.When I use ([svy: tabulate hiv03 v130 , count format(%14.3gc]) the results were different:
four(4) missing information on religion and 4 cases without religion.
Subsequently I got Islam = 7,457 and Christianity=304 when I invoke the svy command, with "if
good" in which case "missing data" or "no religion" were excluded. Table 2 shows Islam=7,242
and Christianity =280 while mine were 7,457 and 304 respectively. A similar difference is occurred
with residence. In table it is Urban= 4,441 and rural=3,098 while mine are 4,564 and 3,205
respectively. Where did I go wrong?
Looking forward to hearing from you.
Thank you.

Subject: Re: Gambia DHS 2013
Posted by Liz-DHS on Wed, 23 Nov 2016 15:06:13 GMT
View Forum Message <> Reply to Message

Dear User,
A response from Dr. Tom Pullum:
Quote:
Can you repeat what you did, but not using svyset? A footnote such as "2 cases had no
information on religion" generally gives an Unweighted number of cases. It can easily happen that
2 unweighted cases would be 4 weighted cases. By the way the weight is the only component of
svyset that is relevant for this kind of check. Please let me know if you still have an inconsistency.

Subject: Re: Gambia DHS 2013
Posted by Ceesay1 on Tue, 29 Nov 2016 09:46:22 GMT
View Forum Message <> Reply to Message

Thank you. I will check again.

Subject: Re: Gambia DHS 2013
Posted by sngacksowe@yahoo.com on Thu, 03 May 2018 00:33:05 GMT
View Forum Message <> Reply to Message

Hi User,
The following steps may also help.
1. Merge the women and hiv dataset.
2. Merge the men and hiv datasets.
3. Append the new men and womendata sets which were form from the previously merged
datasets above.

I did mine as follows:
clear all
version 13.1
set more off
*To locate the datasets
cd "C:\Users\oneness\Desktop\First year thesis\Sarata\"
*MERGING THE DATASETS
*To call the individual/women's dataset
use "GMIR60FL.DTA", clear
sort v001 v002 v003
generate gender = 2
save "women.dta", replace
*To call the HIV dataset
use "GMAR60FL.DTA", clear
ren hivclust v001
ren hivnumb v002
ren hivline v003
sort v001 v002 v003
save "hiv.dta", replace
*Merge women and HIV datasets
merge 1:1 v001 v002 v003 using "women.dta"
drop if hiv03 == .
*keep if _merge == 3
save "women_hiv", replace
*To call the men's dataset
use "GMMR60FL.DTA", clear
generate gender = 1
ren mv* v*
ren mg1* g1*
ren mcas* cas*
sort v001 v002 v003
save "men.dta", replace
*Merge men and HIV datasets
merge 1:1 v001 v002 v003 using "hiv.dta"
drop if hiv03 == .
*keep if _merge == 3
save "men_hiv", replace
*To append the women and men data sets.
use "men_hiv.dta", clear

append using "women_hiv.dta"
save "men_women_hiv.dta", replace

Subject: Re: Gambia DHS 2013
Posted by sngacksowe@yahoo.com on Thu, 03 May 2018 00:39:57 GMT
View Forum Message <> Reply to Message

Hi
I am using The Gambia DHS. The svy command is for weighted samples which differs from the
unweighted samples. Probably you used the unweighted which cause the difference.
I have my own issues too. I have posted here for help but no reply yet. Kindly lend me a helping
hand if you can. I have problems in coding the age difference between partners.As some men
have more than one wife, it was difficult for me to code age difference between partners so as to
be able to assess age-disparate relationship. Please help. Thanks

Subject: Re: Gambia DHS 2013
Posted by Liz-DHS on Thu, 03 May 2018 18:59:47 GMT
View Forum Message <> Reply to Message

A response from senior technical specialist, Dr. Tom Pullum,
Quote:
I strongly recommend that you use the CR file, which matches men and women partners on the
same record. The age difference (man minus woman) is mv012-v012. You would weight with
mv005, the man's weight. Non response is generally higher for men than for women, so we
recommend using the man's weight. There has been some recent work by Stan Becker proposing
an alternative weight.

